WILTRON

Programming the Model 360 Vector Network Analyzer Using HP Basic

January 1990

Network Analysis Division • 490 Jarvis Drive • Morgan Hill, CA 95037-2809 • Tel. (408) 778-2000 • Telex 285227 WILTRON MH • FAX (408) 778-0239

	S11	S22	S21		S12	
FREQUENCY (GHz)	SWR (Units)	SWR (Units)	MAG (dB)	PHASE (Deg)	MAG (dB)	PHASE (Deg)
1.00	1.05	1.11	-.32	-.25	-.09	-.12
1.10	1.05	1.06	-.34	-.42	-.09	-.22
1.20	1.06	1.14	-.35	-.17	-.08	-.22
1.30	1.07			14	-.08	-.34
1.40	1.07				-.10	-.14
1.50	1.07				-.12	-.23
1.60	1.07				-.11	-.35
1.70	1.06				-.13	-.25
1.80	1.05				-.13	-.12
1.90	1.04				-.16	.02
2.00	1.04				-.06	-.51
2.10	1.04				-.08	-.22
2.20	1.05				-.12	-.57
2.30	1.06				-.09	-.15
	1.07				-.30	-1.18
	.09				-.14	-.76
	.10				-.12	1.63
	.12	1.			25	.38
	.12	1.2				-.29
	.11	1.16				.16
	.11	1.38				-1.83
	10	1.40		-.95	-.08	-.07
		1.45	-.53	.32	-.30	-.43
	.09	1.29	-.30	.55	-.13	-.31
	.09	1.23	-.29	-.07	-.22	.50
	.10	1.13	-.35	-.47	-.15	-.04
	.11	1.09	-.39	-.20	-.18	.09
	.12	1.13	-.37	.01	-.17	-.11
	.14	1.19	-.37	-.09	-.11	-.17
	.15	1.16	-.34	.01	-.15	-.18





The 360 Vector Network Analyzer has complete GPIB capability and is easy to program. It can be controlled by any computer with a GPIB interface and is readily integrated into automatic measurement systems.

This application note will focus on programs written in HP Basic to run on HP controllers. Knowledge of HP Basic is assumed. A basic measurement program will be developed. It will consist of a common main program and independent subprograms. Each will include listings that illustrate the ease with which the Wiltron 360 can be programmed to do common operations. It will be easy for the user to adapt these operations to his own needs.

Getting Started

A typical set up would be connected as shown in Figure 1. A few characteristics of 360 GPIB implementation should be noted and understood. These are explained below.

Mnemonics

All 360 mnemonics are 3 character alphanumerics, related in most cases to the function to be performed. For example RST (Reset) and SRT 2 GHz (Start frequency = 2 GHz). Although there are more than 300 mnemonics that reflect the overall capability of the 360 to work in specialized applications, only a small number are used in most ATE programs.

Strings

The 360 will accept commands in the form of strings. Spaces between mnemonics are not necessary, but they may be used to separate individual mnemonics. For example, the mnemonic string:

"RST SRT 2 GHZ STP 18 GHZ MK1 10 GHZ AVG 5 XX1"

Performs the following: The 360 is reset to default parameters, a frequency range from 2 to 18 GHz is established, Marker 1 is set to 10 GHz, and Averaging is set to 5.

The sequence of commands within a string is often important. For example, a marker must be established before the MMX (Marker to Max) function is used.

Numeric Entries

All numeric entries require a terminator, such as DBL (dB), GHZ (GHz) and XX1 (x1). Note that even unitless quantities require a terminator. For example, numerics such as 5 averages must use the unitless terminator XX1 (x1).

Additional Information

The 360 Operation Manual contains a complete description of the GPIB capability including a complete list of mnemonics and additional examples.



Figure 1. *Test setup for a 360 Vector Network Analyzer system with a GPIB controller*

Program Development

This sample program automates 360 Vector Network Analyzer calibration and device measurements. It consists of a main program and subprograms (drivers). The main program establishes the GPIB conditions. The subprograms perform specific functions and are called by the main program. The subprograms perform the following functions:

Calibration
Front Panel Setup
Measurement and Data Output.

```
10    !           PROGRAM TO AUTOMATE CALIBRATION AND MEASUREMENT
20    !              ON A WILTRON 360 VECTOR NETWORK ANALYZER
30    !       USING AN HP9000 SERIES 200 OR 300 CONTROLLER WITH GPIB
40    !                          WRITTEN IN HP BASIC
50    !
60    !                          ---MAIN PROGRAM---
70    !  ASSIGN I/O PATH FOR GPIB AND 360, SET 360 TO REMOTE OPERATION,
80    !                      CLEAR GPIB, AND RESET 360.
90    !
100   ASSIGN @Gpib TO 7
110   ASSIGN @Na TO 706
120   ASSIGN @Na1 TO 706;FORMAT OFF
130   REMOTE @Gpib
140   ABORT @Gpib
150   CLEAR @Na
160   !
170   !      CALL SUBPROGRAMS TO PERFORM CALIBRATION, FRONT PANEL SETUP,
180   !                        MEASUREMENT AND DATA OUTPUT.
190   !
200   Cal_12_term(@Na)
210   Fp_setup(@Na)
220   Meas_device(@Na,@Na1)
230   !
240   !             RETURN 360 TO LOCAL OPERATION AND END PROGRAM.
250   !
260   LOCAL @Gpib
270   END
280   !
```

Figure 2. *Program for initializing the 360 and GPIB, performing subprograms, and returning 360 to local operation*

```
290   !                    ---CALIBRATION SUBPROGRAM---
300   !
310   SUB Cal_12_term(@Na)
320   PRINT TABXY(35,1);"CALIBRATION"
330   OUTPUT @Na;"LTC SCM C12 DFC FRS 1 GHZ FRI 100 MHZ FRP 41 XX1 FIL DFD P1C
      CFK P2C CMK BBL BEG ONP"
340   ENTER @Na;N$
350   DISP "CONNECT BROADBAND LOADS TO PORT 1 AND PORT 2"
360   GOSUB Continue
370   DISP "CONNECT OPEN TO PORT 1 AND SHORT TO PORT 2"
380   GOSUB Continue
390   DISP "CONNECT SHORT TO PORT 1 AND OPEN TO PORT 2"
400   GOSUB Continue
410   DISP "CONNECT PORT 1 TO PORT 2"
420   GOSUB Continue
430   DISP "12 TERM CALIBRATION COMPLETE"
440   SUBEXIT
450 Continue: BEEP
460   LOOP
470   ON KEY 5 LABEL "TAKE CAL DATA" GOTO 490
480   END LOOP
490   DISP "TAKING CALIBRATION DATA..."
500   OUTPUT @Na;"TCD NCS ONP"
510   ENTER @Na;N$
520   RETURN
530   SUBEND
540   !
```

Figure 3. *Subprogram for performing a 12-term calibration over the GPIB*

Main Program

The main program shown in Figure 2 initializes the 360 and GPIB, performs the subprograms and returns the 360 to local operation.

First, the 360 and GPIB are initialized to a known state so that the program starts under identical conditions each time it is run.

The ASSIGN command establishes as I/O path for the HP controller to 7 and for the 360 to 706 (this assumes the 360 is internally set to address 6 on the 360 GPIB). FORMAT OFF is used when data is to be transferred in binary floating point.

The REMOTE command places the 360 and other instruments on the bus into the remote state.

The ABORT command clears and takes control of the bus.

The CLEAR command, as implemented on the 360, clears the bus and resets 360 to default parameters. It should be used before 360 parameters are established by the program.

Next, subprograms are sequentially called to perform specific functions such as calibration, front panel setup, device measurement and data output.

Finally, the LOCAL command instructs the HP controller to return the 360 to local operation and the program is ended.

Subprograms

Calibration

The calibration subprogram shown in Figure 3 performs a 12-term calibration at discrete frequencies from 1 to 5 GHz with 100 MHz spacing, for a total of 41 frequencies. Line 330 uses the OUTPUT command to send the 360 a mnemonic string that establishs the parameters for the calibration to be performed. Port 1 is a female K Connector® and Port 2 is a male K Connector. A broadband termination is used.

The controller determines if the 360 is ready for the next step of the calibration sequence by requesting "dummy" data output from the 360 after executing NCS. In this example, ONP instructs the 360 to output the number of points when the calibration step is complete. The controller reads this data from the 360 into a string N$ using the ENTER command and continues to the next step.

Front Panel Setup

The front panel setup subprogram shown in Figure 4 creates a 360 display with the proper format, enhancements, and markers to enable the operator to observe the device performance.

A four channel display is set up. S11 and S22 are displayed as SWR. S21 and S12 are displayed as Log Magnitude and Phase. I.F. Bandwidth is set to MINIMUM (100 Hz). Averaging is set to 10. Marker 1 is set to 2.7 GHz and Marker 2 is set to 4.5 GHz.

```
550   !                ---FRONT PANEL SETUP PROGRAM---
560   !
570   SUB Fp_setup(@Na)
580   PRINT TABXY(32,1);"FRONT PANEL SETUP"
590   OUTPUT @Na;"D14 CH1 S11 SWR CH2 S22 SWR CH3 S21 MPH CH4 S12 MPH"
600   OUTPUT @Na;"IFM AVG 10 XX1"
610   OUTPUT @Na;"MK1 2.7 GHZ MK2 4.5 GHZ ONP"
620   ENTER @Na;N$
630   SUBEND
640   !
```

Figure 4. *Subprogram for a performing a front panel setup over the GPIB*

```
650   !          ---MEASUREMENT AND DATA OUTPUT PROGRAM---
660   !
670   SUB Meas_device(@Na,@Na1)
680   OPTION BASE 1
690   DIM Freq(41,1),S11(41,2),S22(41,2),S21(41,2),S12(41,2)
700   INTEGER Preamble,Size
710   PRINT TABXY(27,1);"MEASUREMENT AND DATA OUPUT"
720   DISP "Connect Device Under Test and press 'MEASURE' when ready"
730   BEEP
740   LOOP
750   ON KEY 5 LABEL "MEASURE" GOTO 770
760   END LOOP
770   DISP "Taking measurement data..."
780   OUTPUT @Na;"TRS WFS HLD PFS"
790   OUTPUT @Na;"FMB MSB OFV CH1 OFD CH2 OFD CH3 RDA OFD CH4 RDA OFD"
800   ENTER @Na1;Preamble,Size,Freq(*)
810   ENTER @Na1;Preamble,Size,S11(*)
820   ENTER @Na1;Preamble,Size,S22(*)
830   ENTER @Na1;Preamble,Size,S21(*)
840   ENTER @Na1;Preamble,Size,S12(*)
850   DISP " "
860   PRINT TABXY(1,4);"             S11       S22          S21                 S12"
870   PRINT "FREQUENCY   SWR       SWR       MAG    PHASE       MAG    PHASE"
880   PRINT "  (GHz)   (Units)  (Units)     (dB)   (Deg)       (dB)   (Deg)"
890   FOR I=1 TO 41 STEP 5
900   PRINT USING "7(4D.2D,2X)";Freq(I,1)/1.E+9,S11(I,1),S22(I,1),S21(I,1),S21(I
,2),S12(I,1),S12(I,2)
910   NEXT I
920   DISP "Select 'ANOTHER DEVICE' or 'QUIT' to end program"
930   LOOP
940   ON KEY 5 LABEL "ANOTHER DEVICE " GOTO 720
950   ON KEY 9 LABEL "QUIT" GOTO 970
960   END LOOP
970   DISP "PROGRAM ENDED"
980   SUBEND
```

Figure 5. *Subprogram for measurement and data output over the GPIB*

Measurement and Data Output

The measurement and data output subprogram as shown in Figure 5 prompts the operator to connect a device and measure it by selecting function key k5.

The 360 is instructed to trigger a sweep, wait a full sweep, and go into hold mode. The 360 is then instructed to output the frequency values and measurement data from all four channels in the IEEE 64 bit binary floating point format, sending the most significant byte first.

The data is read into arrays in the controller, formatted, and then displayed on the controller monitor as shown in Figure 6.

```
              S11       S22          S21                 S12
FREQUENCY     SWR       SWR        MAG     PHASE       MAG     PHASE
  (GHz)     (Units)   (Units)      (dB)    (Deg)       (dB)    (Deg)
   1.00       1.05      1.04       -.09      -.30       -.07      -.39
   1.50       1.06      1.06       -.10      -.35       -.10      -.40
   2.00       1.04      1.05       -.12      -.37       -.11      -.39
   2.50       1.08      1.08       -.13      -.41       -.13      -.46
   3.00       1.11      1.12       -.15      -.38       -.14      -.34
   3.50       1.11      1.11       -.16      -.37       -.16      -.40
   4.00       1.12      1.12       -.17      -.32       -.16      -.45
   4.50       1.14      1.14       -.19      -.40       -.20      -.38
   5.00       1.19      1.19       -.22      -.30       -.20      -.39
```

Figure 6. *Measurement data as displayed on the controller*

Printed in USA
January 1990

WILTRON

P/N: 11410-00039
Rev A